Métodos Computacionais Numéricos e Algébricos com Maple

Revisão de Cálculo

Fernando Deeke Sasse
CCT - UDESC
fernandodeeke@gmail.com

Teoremas Fundamentais

1. **Teorema de Rolle.** Suponha que f é uma função contínua no intervalo $[a, b]$ e diferenciável em (a, b). Se $f(a) = f(b)$, e diferenciável em (a, b) $f'(c) = 0$. Então existe ao menos um ponto neste intervalo onde $f'(x_0) = 0$.

Exemplo. Tomemos $f = \sin(x)$ e o intervalo $[1, 1 + 2\pi]$. Certamente a função tem valores iguais nos dois extremos e existe ao menos um ponto neste intervalo, onde $f'(c) = 0$. Isso pode ser comprovado através do comando **RollesTheorem**, como mostrado abaixo:

```
> with(Student[Calculus1]):
> RollesTheorem(sin(x), x=1..1+2*Pi);
```

Illustration of Rolle's Theorem



For the function $f(x) = \sin(x)$ on the interval $[1, 1 + 2\pi]$, a graph showing $f(x)$, the line connecting $(1, f(1))$ and $(1 + 2\pi, f(1 + 2\pi))$, tangents parallel to the line connecting $(1, f(1))$ and $(1 + 2\pi, f(1 + 2\pi))$.

2. **Teorema do Valor Médio**. Esta é uma generalização do Teorema de Rolle que estabelece que, se f é contínua em $[a, b]$ e diferenciável em (a, b), então existe um ponto c em (a, b) tal que

$$f'(c) = \frac{f(b) - f(a)}{b - a}.$$

Exemplo. Tomemos novamente $f = \sin(x)$ e o intervalo $[1, 3]$. Certamente a função tem valores iguais nos dois extremos e existe ao menos um ponto neste intervalo, onde

$$f'(c) = \frac{f(3) - f(1)}{2 - 1} = \frac{(\sin(3) - \sin(1))}{1} = -0.7003509767.$$

```
> MeanValueTheorem(sin(x), x=1..3);
```

3. **Teorema do Valor Extremo**. Se f é uma função contínua e limitada no intervalo $[a, b]$, então f atinge seu valor máximo e seu valor mínimo ao menos uma vez. Ou seja, existem $c_1, c_2 \in [a, b]$ tal que

$$f(c_1) \leq f(x) \leq f(c_2)$$

para todo $x \in [a, b]$. Além disso, se f for diferenciável em (a, b), então os números c_1 e c_2 ocorrem nas extremidades de $[a, b]$ ou nos pontos para os quais f' é zero.

Exemplo. Determine

$$\max_{0 \leq x \leq 5} |f(x)|$$

para

$$f(x) = 3\sin(3x) - \cos(x)\,x + 2$$

Resolvamos este problema com Maple.

```
> f:=3*sin(3*x)-cos(x)*x+2;
```

$$f := 3\sin(3x) - \cos(x)\,x + 2 \quad \textbf{(1.1)}$$

Façamos um gráfico para termos uma noção da localização dos máximos

```
> plot(f,x=0..5);
```



Certamente temos um máximo absoluto entre os pontos 2 e 3. Para determiná-lo com precisão máximo calculemos a derivada de f:

```
> df:=diff(f,x);
```

$$df := 9\cos(3x) + \sin(x)\,x - \cos(x) \quad \textbf{(1.2)}$$

O máximo corresponde a um zero desta função entre 2 e 3:

```
> xm:=fsolve(df=0,x=2..3);
```

$$xm := 2.695185292 \quad \textbf{(1.3)}$$

Exercício Determine
(a) min $_{0 \le x \le 5} |f(x)|$,
(b) max$_{0 \le x \le 30} |f(x)|$.

4. **Integral de Riemann.** A integral de Riemann da função f no intervalo $[a, b]$ é definida através do seguinte limite, caso ele exista:

$$\int_a^b f(x)\, dx = \lim_{\Delta x_i \to 0} \sum_{i=1}^{n} f(\zeta_i) \Delta x_i ,$$

sendo que os números x_0, x_1 ..., x_n satisfazem a $a = x_0 \le x_1 \le \cdots \le x_n = b$, sendo $\Delta x_i = x_i - x_{i-1}$ para $i = 1, \ldots, n$ e $\zeta_i \in [x_{i-1}, x_i]$.

Suponhamos que f é uma função contínua em um intervalo $[a, b]$. Escolhendo intervalos gualmente espaçados,

$$\Delta x_i = \Delta x = \frac{(b-a)}{n},$$

e escolhendo $\zeta_i = x_i$, temos

$$\int_a^b f(x)\, dx = \lim_{\Delta x_i \to 0} \frac{(b-a)}{n} \sum_{i=1}^{n} f(x_i)$$

Notemos que

$$x_i = a + i\,\Delta x = a + i\,\frac{(b-a)}{n} .$$

Na fórmula acima tomamos o limite da chamada ***soma de Riemann superior***, pois avaliamos f sempre no extremo superior do intervalo. A soma de Riemann inferior é obtida definindo $\zeta_i = x_{i-1}$, enquanto que a soma de Riemann de ponto médio é obtida definindo $\zeta_i = \frac{x_i + x_{i-1}}{2}$.

Exemplo. Calculemos uma aproximação para a integral de $f(x) = \sin(x)$ no intervalo $[1, 2]$, usando a soma de Riemann de ponto médio, com 12 intervalos.

```
> a:=0.;b:=2.;n:=12;
```

$$a := 0.$$
$$b := 2.$$
$$n := 12 \qquad \textbf{(1.4)}$$

```
> Delta:=(b-a)/n;
```

$$\Delta := 0.1666666667 \qquad \textbf{(1.5)}$$

```
> f:=x->sin(x);
```

$$f := x \to \sin(x) \qquad \textbf{(1.6)}$$

Geramos agora o que chamamos de ***malha de pontos***, ou seja, o conjunto de pontos onde f é avaliada:

```
> z:=[seq(a+j*Delta,j=0..n)];
```

z := [0., 0.1666666667, 0.3333333334, 0.5000000001, 0.6666666668, 0.8333333335, 1.000000000, 1.166666667, 1.333333334, 1.500000000, 1.666666667, 1.833333334, 2.000000000] **(1.7)**

A soma de Riemann é então dada por

```
> I1 := Delta*(sum(f((z[i]+z[i+1])*(1/2)), i = 1 .. n));
```
$$I1 := 1.417787225 \tag{1.8}$$

O valor real da integral é dado por

```
> I1r:=evalf(int(sin(x),x=0..2));
```
$$I1r := 1.416146836 \tag{1.9}$$

O erro relativo cometido é dado por

```
> Erro:=abs((I1r-I1)/I1r);
```
$$Erro := 0.001158346690 \tag{1.10}$$

Podemos visualizar a soma de Riemann através do comando:

```
> ApproximateInt(f(x), 0..2, 'partition' = 12, 'method' =
midpoint, 'partitiontype' = subintervals, 'output' = 'plot',
'boxoptions' = ['filled' = ['transparency'=.5]]);
```



A midpoint Riemann sum approximation of $\int_0^2 f(x)\, dx$, where $f(x) = \sin(x)$ and the partition is uniform. The approximate value of the integral is 1.417787224. Number of subintervals used: 12.

Exercício Determine uma aproximação para $\int_0^2 \frac{1}{\sqrt{x^3+1}}\, dx$, com 15 intervalos, usando a soma de Riemann inferior, superior e de ponto médio. Compare os resultados com a aquele dado pelo comando `int` do Maple. Visualize as somas integrais usando o comando `ApproximateInt` do Maple.

5. **Teorema do Valor Médio para Integrais.** Se f e g são funções contínuas no intervalo $[a, b]$ e $g(x) \geq 0$, então existe um número $c \in [a, b]$ tal que

$$\int_a^b f(x)\, g(x)\, \mathrm{d}x = f(c) \int_a^b g(x)\, \mathrm{d}x.$$

Quando $g(x) \equiv 1$ temos o usual Teorema do Valor Médio para Integrais, que fornece o valor médio de f no intervalo $[a, b]$:

$$f(c) = \frac{1}{b-a} \int_a^b f(x)\, \mathrm{d}x.$$

Exercício. Dados $f(x) = \cos(x)$ e $g(x) = x^2 + x - 1$, determine $c \in [0, 5]$ tal que o Teorema do Valor Médio para Integrais seja satisfeito.

6. **Teorema de Rolle Generalizado.** Seja f uma função contínua e limitada no intervalo $[a, b]$ e n vezes diferenciável em (a, b). Se $f(x) = 0$ em $n+1$ pontos distintos $x_0, x_1, \ldots, x_n \in [a, b]$, então existe um número $c \in (a, b)$ tal que $f^{(n)}(c) = 0$.

Exemplo. Seja $f = x \cdot \cos(x) - x^2 \exp(-x)$. Consideremos o intervalo $[0, 5]$. Plotemos esta função no intervalo de interesse.

```
> f:=x^2-x*sin(4*x)-2*x-2;
```

$$f := x^2 - x \sin(4x) - 2x - 2 \tag{1.11}$$

```
> plot(f,x=1..4);
```



Neste intervalo temos $n = 3$ raízes. Então a terceira derivada de f é dada por

```
> df3:=diff(f,x$3);
```

$$df3 := 48 \sin(4x) + 64x \cos(4x) \tag{1.12}$$

Façamos o gráfico desta função no intervalo de interesse:

```
> plot(df3,x=1..4);
```



Determinemos a primeira raiz:

```
> r:=fsolve(df3,x=1..2);
```

$$r := 1.308234613 \tag{1.13}$$

Portanto, podemos escolher $c = 1.308234613$.

7. **Teorema do Valor Intermediário.** Sejam f uma função contínua e limitada no intervalo $[a, b]$ e K um número qualquer entre $f(a)$ e $f(b)$. Então existe um número $c \in (a, b)$ para o qual $f(c) = K$.

Exemplo. Seja a função

```
> f:=x->exp(-x)+sin(x)-x/6;
```

$$f := x \rightarrow \mathrm{e}^{-x} + \sin(x) - \frac{1}{6}\, x \tag{1.14}$$

Como

```
> f(2.);f(4.);
```

$$0.7112993767$$
$$-1.405153523 \tag{1.15}$$

ou seja, $f(2) = 0.7113$ e $f(4) = -1.4051$, o Teorema do Valor Intermediário implica que para $K = 0 \in [f(2), f(4)]$ deve existir c (a raiz de f) tal que $f(c) = 0$. De fato, c é dado por

```
> c:=fsolve(f(x),x=2..4);
```

$$c := 2.738947863 \tag{1.16}$$

8. **Teorema de Taylor.** Suponhamos que $f \in C^n[a, b]$ (ou seja, f admite derivada contínua de ordem até n), que $f^{(n+1)}$ exista em $[a, b]$ e $x_0 \in [a, b]$. Então para todo $x \in [a, b]$ existe um número $\xi(x) \in [x_0, x]$ tal que

$$f(x) = P_n(x) + R_n(x),$$

com

$$P_n(x) = f(x_0) + f'(x_0)(x - x_0) + \frac{f''(x_0)}{2!}(x - x_0)^2 + \cdots + \frac{f^{(n)}(x_0)}{n!}(x - x_0)^n,$$

$$= \sum_{k=0}^{n} \frac{f^{(k)}(x_0)}{k!} (x - x_0)^k .$$

e

$$R_n(x) = \frac{f^{(n+1)}[\xi(x)]}{(n+1)!} (x - x_0)^{n+1} .$$

Aqui $P_n(x)$ é denominado n-ésimo polinômio de Taylor para f em torno de x_0, e $R_n(x)$ é o erro de truncamento ou resto, associado a $P_n(x)$. No limite $n \to \infty$ $P_n(x)$ é denominado série de Taylor de $f(x)$ em torno de x_0.

Notemos que o número $\xi(x)$ é desconhecido. O melhor que podemos fazer na prática é determinar um limite superior para $f^{(n+1)}[\xi(x)]$ num dado intervalo. Uma estimativa para o erro na aproximação de $f(x)$ por $P_n(x)$ é dado por

$$E_n = |f(x) - P_n(x)| = \max_{\xi(x) \in [x_0, x]} \left| \frac{f^{(n+1)}[\xi(x)]}{(n+1)!} (x - x_0)^{n+1} \right| = \frac{M}{4!} |x - x_0|^{n+1}$$

sendo

$$M := \max_{\xi(x) \in [x_0, x]} |f^{(n+1)}[\xi(x)]| .$$

Exemplo. Seja $f = e^{-x^2} \sin(x)$.
(i) Determinemos os polinômios de Taylor $P_2(x)$ e $P_3(x)$ de f em torno de $x_0 = 0$:

```
> f:=exp(-x^2)*sin(x);
```

$$f := e^{-x^2} \sin(x) \tag{1.17}$$

```
> s2:=series(f,x=0,3);
```

$$s2 := x + O(x^3) \tag{1.18}$$

O símbolo $O(x^3)$ representa um polinômio de grau igual ou superior a 3. Devemos converter a expressão à forma polinomial:

```
> P2:=convert(s2,polynom);
```

$$P2 := x \tag{1.19}$$

Por outro lado,

```
> s3:=series(f,x=0,4);
```

$$s3 := x - \frac{7}{6} x^3 + O(x^4) \tag{1.20}$$

```
> P3:=convert(s3,polynom);
```

$$P3 := x - \frac{7}{6} x^3 \tag{1.21}$$

Façamos um gráfico de f e de P_3:

```
> plot([f,P3],x=-1..1,color=[blue,red],legend=['f',P[3]]);
```



(ii) Façamos uma estimativa do erro cometido na representação de $f(0.5)$ por $P_3(0.5)$. De acordo com o teorema de Taylor,

$$R_3(x) = \frac{f^{(4)}[\xi(x)]}{4!}\,x^4 .$$

Um limitante superior para a magnitude de $R_3(x)$ é dada por

$$|R_3(x)| \approx \frac{M}{4!}\,x^4 ,$$

sendo

$$M := \max_{\xi \in [0,\,0.5]} \left| f^{(4)}[\xi(x)] \right| .$$

Calculemos $f^{(4)}(x)$:

```
> df4:=diff(f,x$4);
```

$$df4 := 25\,\mathrm{e}^{-x^2}\sin(x) - 72\,x^2\,\mathrm{e}^{-x^2}\sin(x) + 56\,x\,\mathrm{e}^{-x^2}\cos(x) + 16\,x^4\,\mathrm{e}^{-x^2}\sin(x) - 32\,x^3\,\mathrm{e}^{-x^2}\cos(x) \qquad \textbf{(1.22)}$$

Façamos um gráfico para determinar o a magnitude máxima de $R_3(\xi)$ no intervalo $[0, 0.5]$:

```
> plot(df4,x=0..0.5);
```



Portanto, o valor máximo de $|R_3|$ ocorre em torno de $\xi = 0.4$ e

```
> M:=abs(evalf(subs(x=0.4,df4)));
```

$$M := 20.58293532 \tag{1.23}$$

O limitante para o máximo erro é dado, então, por:

```
> E3:=M/(4!)*(0.5-0)^4;
```

$$E3 := 0.05360139406 \tag{1.24}$$

O erro real é dado por

```
> Ereal:=evalf(subs(x=0.5,abs(f-P3)));
```

$$Ereal := 0.0192103182 \tag{1.25}$$

que é menor que o limitante, como já poderíamos esperar.

Exemplo. A aproximação de Taylor pode ser utilizada para aproximar integrais, como por exemplo,

$$\int_0^{0.5} e^{-x^2} \sin(x)\, dx.$$

Notemos que, de acordo com o Teorema de Taylor, o integrando $f = e^{-x^2} \sin(x)$ pode ser escrito com

$$f(x) = P_3(x) + R_3(x) = x - \frac{7}{6} x^3 + \frac{f^{(4)}[\xi(x)]}{4!} x^4,$$

de modo que

$$\int_0^{0.5} f(x)\, dx = \int_0^{0.5} e^{-x^2} \sin(x)\, dx = \int_0^{0.5} P_3(x)\, dx + \int_0^{0.5} R_3(x)\, dx.$$

A primeira integral pode ser calculada diretamente, mas a segunda pode ser somente estimada, pois $\xi(x)$ é desconhecido. Usamos então a aproximação:

$$\int_0^{0.5} f(x)\, dx = \int_0^{0.5} e^{-x^2} \sin(x)\, dx \approx \int_0^{0.5} P_3(x)\, dx$$

$$= \int_0^{0.5} \left(x - \frac{7}{6} x^3 \right) dx$$

sendo o erro limitado por

$$E = \left| \int_0^{0.5} R_3(x)\, \mathrm{d}x \right| \leqslant \frac{M}{4!} \left| \int_0^{0.5} x^4\, \mathrm{d}x \right|.$$

Façamos os cálculos no Maple:

```
> I3:=int(P3,x=0..0.5);
```

$$I3 := 0.1067708333 \quad \textbf{(1.26)}$$

```
> E:=abs(M*(int(x^4, x = 0 .. .5))/factorial(4));
```

$$E := 0.005360139408 \quad \textbf{(1.27)}$$

O valor da integral, dado pelo melhor algoritmo do Maple é:

```
> IR:=int(f,x=0..0.5);
```

$$IR := 0.1084092900 \quad \textbf{(1.28)}$$

A melhor estimativa para o erro é então dado por

```
> IV:=abs(IR-I3);
```

$$IV := 0.0016384567 \quad \textbf{(1.29)}$$

Exercício. Seja $f(x) = x \exp(-x^2)$.

(a) Determine o polinômio de Taylor $P_7(x)$ e estime o erro cometido na aproximação $f \approx P_7(x)$. Calcule o erro verdadeiro.

(b) Exiba um gráfico comparando $f(x)$ e $P_7(x)$ no intervalo $[-1, 1]$.

(c) Calcule aproximação

$$\int_0^{0.6} f(x)\, \mathrm{d}x \approx \int_0^{0.6} P_7(x)\, \mathrm{d}x$$

(d) Use o Teorema de Taylor para determinar um limite superior para o erro.

(e) Calcule o erro verdadeiro (a integral tem solução exata).